/*

Paasch, Stecker (2020):
When Europe hits the subnational authorities: The
transposition of EU directives in Germany between 1990
and 2018, in: Journal of Public Policy.

*/



version 15
clear
set more off


*** Load data
use "\paasch-stecker_data.dta", clear

*** package requirements (uncomment if necessary)
* net install st0085_2.pkg


*** Model specifications
* Model 1
logit delay_notif_dum i.subnational i.competence ///
		i.preleg deadline_y i.eu_agent ///
		recitals i.polsector, vce(cluster dir_code) or
est store model_1


* Model 2		
logit delay_severe i.subnational i.competence ///
		i.preleg deadline_y i.eu_agent ///
		recitals i.polsector, vce(cluster dir_code) or
est store model_2


* Plot coefficients Model 1+2
coefplot /*Model_1*/ ///
		(model_1, transform(* = min(max(@,0),8)) xlab(0.0 (1) 7) label(Model 1: Delay)) ///
		/*Model_2 - severe delay*/ ///
		(model_2, transform(* = min(max(@,0),8)) xlab(0.0 (1) 7) label(Model 2: Severe delay) legend(pos(6) col(2))), ///
			eform xline(1, lcolor(black)) xtitle(" " "Odds Ratios", size(small)) ///
		drop(*.polsector *.dir_treaty_gr3 _cons) ///
		coeflabel(1.subnational = "{bf:Subnational Implementor}" ///
		1.competence = "{bf:National Responsibility}" ///
		1.preleg = "{bf:Pre-legislation}" ///
		deadline_y = "{bf:Deadline}" 2.eu_agent = "{bf:Council}" 3.eu_agent = "{bf:Council+EP}" ///
		recitals = "{bf:Complexity}") scheme(plotplainblind)

		
* Model 3
logit delay_notif_dum i.consent ///
		if preleg == 0, vce(cluster dir_code) or

		
* Model 4
logit delay_notif_dum i.consent ///
		i.eu_agent deadline_y recitals i.polsector if preleg == 0, vce(cluster dir_code) or


* Model 5
xtlogit delay_notif_dum i.subnational i.competence ///
		i.preleg deadline_y i.eu_agent ///
		recitals i.polsector, i(dir_code_n) intpoints(30) or
